Rules for integrands of the form $(d \operatorname{Sec}[e + f x])^n (a + b \operatorname{Sec}[e + f x])^m$

1: $\int (a + b \operatorname{Sec}[e + f x]) (d \operatorname{Sec}[e + f x])^n \, dx$

Derivation: Algebraic expansion

Basis: $a + b z == a + \frac{b}{d} (d z)$

Rule:

$$\int (a + b \operatorname{Sec}[e + f x]) (d \operatorname{Sec}[e + f x])^n \, dx \longrightarrow a \int (d \operatorname{Sec}[e + f x])^n \, dx + \frac{b}{d} \int (d \operatorname{Sec}[e + f x])^{n+1} \, dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])*(d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  a*Int[(d*Csc[e+f*x])^n,x] + b/d*Int[(d*Csc[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,n},x]
```

2: $\int (a + b \operatorname{Sec}[e + f x])^2 (d \operatorname{Sec}[e + f x])^n \, dx$

Derivation: Algebraic expansion

Basis: $(a + b z)^2 == 2 a b z + a^2 + b^2 z^2$

Rule:

$$\int (a + b \operatorname{Sec}[e + f x])^2 (d \operatorname{Sec}[e + f x])^n \, dx \longrightarrow \frac{2 a b}{d} \int (d \operatorname{Sec}[e + f x])^{n+1} \, dx + \int (d \operatorname{Sec}[e + f x])^n \left(a^2 + b^2 \operatorname{Sec}[e + f x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^2*(d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  2*a*b/d*Int[(d*Csc[e+f*x])^(n+1),x] + Int[(d*Csc[e+f*x])^n*(a^2+b^2*Csc[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,n},x]
```

3: $\int \frac{\text{Sec}[e+f x]^2}{a+b\,\text{Sec}[e+f x]}\,dx$

Derivation: Algebraic expansion

Basis: $\frac{z}{a+b z} == \frac{1}{b} - \frac{a}{b\,(a+b z)}$

Rule:

$$\int \frac{\text{Sec}[e+f x]^2}{a+b\,\text{Sec}[e+f x]}\,dx \longrightarrow \frac{1}{b}\int \text{Sec}[e+f x]\,dx - \frac{a}{b}\int \frac{\text{Sec}[e+f x]}{a+b\,\text{Sec}[e+f x]}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  1/b*Int[Csc[e+f*x],x] - a/b*Int[Csc[e+f*x]/(a+b*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f},x]
```

4: $\int \frac{\text{Sec}[e+f x]^3}{a+b\,\text{Sec}[e+f x]}\,dx$

Derivation: Algebraic expansion

Basis: $\frac{z}{a+b z} = \frac{1}{b} - \frac{a}{b\,(a+b z)}$

Rule:

$$\int \frac{\text{Sec}[e+f x]^3}{a+b\,\text{Sec}[e+f x]}\,dx \longrightarrow \frac{\text{Tan}[e+f x]}{b f} - \frac{a}{b}\int \frac{\text{Sec}[e+f x]^2}{a+b\,\text{Sec}[e+f x]}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^3/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -Cot[e+f*x]/(b*f) - a/b*Int[Csc[e+f*x]^2/(a+b*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f},x]
```

5. $\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n \, dx$ when $a^2 - b^2 == 0$

1: $\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n \, dx$ when $a^2 - b^2 == 0 \land m \in \mathbb{Z}^+$

- Derivation: Algebraic expansion

- Rule: If $a^2 - b^2 == 0 \land m \in \mathbb{Z}^+$, then

$$\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n \, dx \longrightarrow \int \text{ExpandTrig}[(a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n, x] \, dx$$

- Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[ExpandTrig[(a+b*csc[e+f*x])^m*(d*csc[e+f*x])^n,x],x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && IGtQ[m,0] && RationalQ[n]
```

2. $\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,dx$ when $a^2-b^2==0$

1. $\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\wedge\,m>0$

1: $\int \text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2==0$

Derivation: Singly degenerate secant recurrence 1b with $A\to c$, $B\to d$, $m\to\frac{1}{2}$, $n\to-1$, $p\to 0$

Rule: If $a^2-b^2==0$, then

$$\int \text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\,dx\;\longrightarrow\;\frac{2\,b\,\text{Tan}[e+f\,x]}{f\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}$$

Program code:

```
Int[csc[e_.+f_.*x_]*Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*b*Cot[e+f*x]/(f*Sqrt[a+b*Csc[e+f*x]]) /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0]
```

2: $\int \mathrm{Sec}[e+f x]\,(a+b\,\mathrm{Sec}[e+f x])^m\,dx$ when $a^2-b^2==0\wedge m>\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 1b with $n\to 0,\ p\to 0$

Rule: If $a^2-b^2==0\wedge m>\frac{1}{2}$, then

$$\int \mathrm{Sec}[e+f x]\,(a+b\,\mathrm{Sec}[e+f x])^m\,dx\longrightarrow$$
$$\frac{b\,\mathrm{Tan}[e+f x]\,(a+b\,\mathrm{Sec}[e+f x])^{m-1}}{f\,m}+\frac{a\,(2m-1)}{m}\int \mathrm{Sec}[e+f x]\,(a+b\,\mathrm{Sec}[e+f x])^{m-1}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -b*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)/(f*m) + a*(2*m-1)/m*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m-1),x] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0] && GtQ[m,1/2] && IntegerQ[2*m]
```

2. $\int \mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,dx$ when $a^2-b^2=0\ \land\ m<0$

1: $\int \frac{\mathrm{Sec}[e+f\,x]}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx$ when $a^2-b^2=0$

- Derivation: Singly degenerate secant recurrence 2a with $A\to 1,\ B\to 0,\ m\to -1,\ n\to 0,\ p\to 0$

- Rule: If $a^2-b^2=0$, then

$$\int \frac{\mathrm{Sec}[e+f\,x]}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx \longrightarrow \frac{\mathrm{Tan}[e+f\,x]}{f\,(b+a\,\mathrm{Sec}[e+f\,x])}$$

- Program code:

```
Int[csc[e_.+f_.*x_]/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -Cot[e+f*x]/(f*(b+a*Csc[e+f*x])) /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0]
```

2: $\int \frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2==0$

Author: Martin on sci.math.symbolic on 10 March 2011

Derivation: Integration by substitution

Basis: If $a^2-b^2==0$, then $\frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}==\frac{2}{f}\,\text{Subst}\left[\frac{1}{2\,a+x^2},\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\right]\partial_x\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}$

Rule: If $a^2-b^2==0$, then

$$\int \frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx\;\rightarrow\;\frac{2}{f}\,\text{Subst}\left[\int\frac{1}{2\,a+x^2}\,dx,\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\right]$$

–

Program code:

```
Int[csc[e_.+f_.*x_]/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2/f*Subst[Int[1/(2*a+x^2),x],x,b*Cot[e+f*x]/Sqrt[a+b*Csc[e+f*x]]] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0]
```

3: $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,dx$ when $a^2-b^2==0\,\land\,m<-\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 2b with $n\to 0,\ p\to 0$

Rule: If $a^2-b^2==0\,\land\,m<-\frac{1}{2}$, then

$$\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,dx\,\longrightarrow$$
$$-\frac{b\,\text{Tan}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m}{a\,f\,(2\,m+1)}+\frac{m+1}{a\,(2\,m+1)}\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^{m+1}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  b*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(a*f*(2*m+1)) + (m+1)/(a*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0] && LtQ[m,-1/2] && IntegerQ[2*m]
```

3. $\int \mathrm{Sec}[e+f\,x]^2\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,dx$ when $a^2-b^2==0$

1: $\int \mathrm{Sec}[e+f\,x]^2\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\land\,m<-\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 2a with $A\to c,\ B\to d,\ n\to 0,\ p\to 0$

Rule: If $a^2-b^2==0\,\land\,m<-\frac{1}{2}$, then

$$\int \mathrm{Sec}[e+f\,x]^2\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,dx\,\longrightarrow\,\frac{\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m}{f\,(2\,m+1)}+\frac{m}{b\,(2\,m+1)}\int \mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(2*m+1)) +
  m/(b*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0] && LtQ[m,-1/2]
```

2: $\int \mathrm{Sec}[e+f x]^2 \, (a+b\,\mathrm{Sec}[e+f x])^m \, dx$ when $a^2-b^2==0 \land m \nless -\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 2c with $A \to c$, $B \to d$, $n \to 0$, $p \to 0$

Rule: If $a^2-b^2==0 \land m \nless -\frac{1}{2}$, then

$$\int \mathrm{Sec}[e+f x]^2 \, (a+b\,\mathrm{Sec}[e+f x])^m \, dx \longrightarrow \frac{\mathrm{Tan}[e+f x] \, (a+b\,\mathrm{Sec}[e+f x])^m}{f\,(m+1)} + \frac{a\,m}{b\,(m+1)} \int \mathrm{Sec}[e+f x] \, (a+b\,\mathrm{Sec}[e+f x])^m \, dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) +
  a*m/(b*(m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m,x] /;
FreeQ[{a,b,e,f,m},x] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]]
```

4. $\int \mathrm{Sec}[e+fx]^3\,(a+b\,\mathrm{Sec}[e+fx])^m\,dx$ when $a^2-b^2=0$

1: $\int \mathrm{Sec}[e+fx]^3\,(a+b\,\mathrm{Sec}[e+fx])^m\,dx$ when $a^2-b^2=0\wedge m<-\frac{1}{2}$

Derivation: ???

Rule: If $a^2-b^2=0\wedge m<-\frac{1}{2}$, then

$$\int \mathrm{Sec}[e+fx]^3\,(a+b\,\mathrm{Sec}[e+fx])^m\,dx\longrightarrow$$

$$-\frac{b\,\mathrm{Tan}[e+fx]\,(a+b\,\mathrm{Sec}[e+fx])^m}{a\,f\,(2m+1)}-\frac{1}{a^2\,(2m+1)}\int \mathrm{Sec}[e+fx]\,(a+b\,\mathrm{Sec}[e+fx])^{m+1}\,(a\,m-b\,(2m+1)\,\mathrm{Sec}[e+fx])\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^3*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  b*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(a*f*(2*m+1)) -
  1/(a^2*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(a*m-b*(2*m+1)*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0] && LtQ[m,-1/2]
```

2: $\int \mathrm{Sec}[e+fx]^3\,(a+b\,\mathrm{Sec}[e+fx])^m\,dx$ when $a^2-b^2=0\wedge m\not<-\frac{1}{2}$

Derivation: Nondegenerate secant recurrence 1b with $A\to a^2$, $B\to 2ab$, $C\to b^2$, $m\to 0$, $p\to 0$

Rule: If $a^2-b^2=0\wedge m\not<-\frac{1}{2}$, then

$$\int Sec[e+fx]^3 \,(a+b\,Sec[e+fx])^m \,dx \longrightarrow$$
$$\frac{Tan[e+fx]\,(a+b\,Sec[e+fx])^{m+1}}{b\,f\,(m+2)}+\frac{1}{b\,(m+2)}\int Sec[e+fx]\,(a+b\,Sec[e+fx])^m\,(b\,(m+1)-a\,Sec[e+fx])\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^3*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m*(b*(m+1)-a*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f,m},x] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]]
```

5. $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0$

1. $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\land n>0$

1. $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2==0$

1: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2==0\land \frac{a\,d}{b}>0$

– Derivation: Integration by substitution

■ Basis: If $a^2-b^2==0\land \frac{a\,d}{b}>0$, then $\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}==\frac{2\,a}{b\,f}\sqrt{\frac{a\,d}{b}}\,\text{Subst}\left[\frac{1}{\sqrt{1+\frac{x^2}{a}}},\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\right]\partial_x\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}$

Rule: If $a^2-b^2==0\land \frac{a\,d}{b}>0$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}\,dx\,\longrightarrow\,\frac{2\,a}{b\,f}\sqrt{\frac{a\,d}{b}}\,\text{Subst}\left[\int\frac{1}{\sqrt{1+\frac{x^2}{a}}}\,dx,\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\right]$$

– Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*a/(b*f)*Sqrt[a*d/b]*Subst[Int[1/Sqrt[1+x^2/a],x],x,b*Cot[e+f*x]/Sqrt[a+b*Csc[e+f*x]]] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && GtQ[a*d/b,0]
```

2: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2==0\,\land\,\frac{a\,d}{b}\not>0$

Derivation: Integration by substitution

Basis: If $a^2-b^2==0$, then $\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}==\frac{2\,b\,d}{f}\,\text{Subst}\left[\frac{1}{b-d\,x^2},\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}}\right]\partial_x\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}}$

Rule: If $a^2-b^2==0\,\land\,\frac{a\,d}{b}\not>0$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}\,dx\,\rightarrow\,\frac{2\,b\,d}{f}\,\text{Subst}\left[\int\frac{1}{b-d\,x^2}\,dx,\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*b*d/f*Subst[Int[1/(b-d*x^2),x],x,b*Cot[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[d*Csc[e+f*x]])] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && Not[GtQ[a*d/b,0]]
```

2: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\ (d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\ \land\ n>1$

Derivation: Singly degenerate secant recurrence 1b with $A\to c,\ B\to d,\ m\to \frac{1}{2},\ n\to n-1,\ p\to 0$ and algebraic simplification

Rule: If $a^2-b^2==0\ \land\ n>1$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\ (d\,\text{Sec}[e+f\,x])^n\,dx\longrightarrow$$
$$\frac{2\,b\,d\,\text{Tan}[e+f\,x]\ (d\,\text{Sec}[e+f\,x])^{n-1}}{f\ (2\,n-1)\ \sqrt{a+b\,\text{Sec}[e+f\,x]}}+\frac{2\,a\,d\ (n-1)}{b\ (2\,n-1)}\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\ (d\,\text{Sec}[e+f\,x])^{n-1}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -2*b*d*Cot[e+f*x]*(d*Csc[e+f*x])^(n-1)/(f*(2*n-1)*Sqrt[a+b*Csc[e+f*x]]) +
  2*a*d*(n-1)/(b*(2*n-1))*Int[Sqrt[a+b*Csc[e+f*x]]*(d*Csc[e+f*x])^(n-1),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && GtQ[n,1] && IntegerQ[2*n]
```

2. $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\ (d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\ \land\ n<0$

1: $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{d\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2==0$

Derivation: Singly degenerate secant recurrence 1a with $A\to 1,\ B\to 0,\ m\to \frac{1}{2},\ n\to -\frac{3}{2},\ p\to 0$

Derivation: Singly degenerate secant recurrence 1c with $A\to a,\ B\to b,\ m\to -\frac{1}{2},\ n\to -\frac{3}{2},\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\ \land\ a^2-b^2==0\ \land\ c^2-d^2\neq 0$, then

$$\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{d\,\text{Sec}[e+f\,x]}}\,dx \to \frac{2\,a\,\text{Tan}[e+f\,x]}{f\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}}$$

–

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]/Sqrt[d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*a*Cot[e+f*x]/(f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[d*Csc[e+f*x]]) /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0]
```

2: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2=0 \land n<-\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 1c with $A\to a,\ B\to b,\ m\to -\frac{1}{2},\ p\to 0$ and algebraic simplification

–

Rule: If $a^2-b^2=0 \land n<-\frac{1}{2}$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(d\,\text{Sec}[e+f\,x])^n\,dx \to$$
$$-\frac{a\,\text{Tan}[e+f\,x]\,(d\,\text{Sec}[e+f\,x])^n}{f\,n\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}+\frac{a\,(2\,n+1)}{2\,b\,d\,n}\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(d\,\text{Sec}[e+f\,x])^{n+1}\,dx$$

–

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  a*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*n*Sqrt[a+b*Csc[e+f*x]]) +
  a*(2*n+1)/(2*b*d*n)*Int[Sqrt[a+b*Csc[e+f*x]]*(d*Csc[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && LtQ[n,-1/2] && IntegerQ[2*n]
```

3: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2=0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2 - b^2 == 0$, then $\partial_x \frac{\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{a-b\,\text{Sec}[e+f x]}} == 0$

Basis: If $a^2 - b^2 == 0$, then $-\frac{a^2\,\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{a-b\,\text{Sec}[e+f x]}}\;\frac{\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{a-b\,\text{Sec}[e+f x]}} == 1$

Basis: $\text{Tan}[e+f x]\,F[\text{Sec}[e+f x]] == \frac{1}{f}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,\text{Sec}[e+f x]\right]\partial_x \text{Sec}[e+f x]$

Rule: If $a^2 - b^2 == 0$, then

$$\int\sqrt{a+b\,\text{Sec}[e+f x]}\,(d\,\text{Sec}[e+f x])^n\,dx \to -\frac{a^2\,\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{a-b\,\text{Sec}[e+f x]}}\int\frac{\text{Tan}[e+f x]\,(d\,\text{Sec}[e+f x])^n}{\sqrt{a-b\,\text{Sec}[e+f x]}}\,dx$$

$$\to -\frac{a^2\,d\,\text{Tan}[e+f x]}{f\,\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{a-b\,\text{Sec}[e+f x]}}\,\text{Subst}\left[\int\frac{(d\,x)^{n-1}}{\sqrt{a-b\,x}}\,dx,\,x,\,\text{Sec}[e+f x]\right]$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  a^2*d*Cot[e+f*x]/(f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[a-b*Csc[e+f*x]])*Subst[Int[(d*x)^(n-1)/Sqrt[a-b*x],x],x,Csc[e+f*x]] /;
FreeQ[{a,b,d,e,f,n},x] && EqQ[a^2-b^2,0]
```

6. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\land\,m+n==0\,\land\,2\,m\in\mathbb{Z}$

1. $\int\frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2==0$

1: $\int\frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2==0\,\land\,d==\frac{a}{b}\,\land\,a>0$

Derivation: Integration by substitution

Basis: If $a^2-b^2==0\,\land\,d==\frac{a}{b}\,\land\,a>0$, then $\frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}==\frac{\sqrt{2}\,\sqrt{a}}{b\,f}\,\text{Subst}\left[\frac{1}{\sqrt{1+x^2}},\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{a+b\,\text{Sec}[e+f\,x]}\right]\,\partial_x\frac{b\,\text{Tan}[e+f\,x]}{a+b\,\text{Sec}[e+f\,x]}$

Rule: If $a^2-b^2==0\,\land\,d==\frac{a}{b}\,\land\,a>0$, then

$$\int\frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx\,\rightarrow\,\frac{\sqrt{2}\,\sqrt{a}}{b\,f}\,\text{Subst}\left[\int\frac{1}{\sqrt{1+x^2}}\,dx,\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{a+b\,\text{Sec}[e+f\,x]}\right]$$

Program code:

```
Int[Sqrt[d_.*csc[e_.+f_.*x_]]/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -Sqrt[2]*Sqrt[a]/(b*f)*Subst[Int[1/Sqrt[1+x^2],x],x,b*Cot[e+f*x]/(a+b*Csc[e+f*x])] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && EqQ[d-a/b,0] && GtQ[a,0]
```

2: $\int \frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2==0$

- Derivation: Integration by substitution

- Basis: If $a^2-b^2==0$, then $\frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}==\frac{2\,b\,d}{a\,f}\,\text{Subst}\left[\frac{1}{2\,b-d\,x^2},\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}}\right]\partial_x\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}}$

- Rule: If $a^2-b^2==0$, then

$$\int \frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx\;\rightarrow\;\frac{2\,b\,d}{a\,f}\,\text{Subst}\left[\int\frac{1}{2\,b-d\,x^2}\,dx,\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}}\right]$$

- Program code:

```
Int[Sqrt[d_.*csc[e_.+f_.*x_]]/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*b*d/(a*f)*Subst[Int[1/(2*b-d*x^2),x],x,b*Cot[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[d*Csc[e+f*x]])] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\wedge\,m+n==0\,\wedge\,m>\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 1a with $A\to 1,\ B\to 0,\ m\to -n-1,\ p\to 0$

Rule: If $a^2-b^2==0\,\wedge\,m+n==0\,\wedge\,m>\frac{1}{2}$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\longrightarrow$$

$$\frac{a\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-1}\,(d\,\text{Sec}[e+f\,x])^n}{f\,m}+\frac{b\,(2\,m-1)}{d\,m}\int (a+b\,\text{Sec}[e+f\,x])^{m-1}\,(d\,\text{Sec}[e+f\,x])^{n+1}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -a*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n/(f*m) +
  b*(2*m-1)/(d*m)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && EqQ[m+n,0] && GtQ[m,1/2] && IntegerQ[2*m]
```

3: $\int (a + b\,\mathrm{Sec}[e + f\,x])^m\,(d\,\mathrm{Sec}[e + f\,x])^n\,dx$ when $a^2 - b^2 == 0 \land m + n == 0 \land m < -\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 2b with $A \to c,\ B \to d,\ n \to -m - 2,\ p \to 0$

Rule: If $a^2 - b^2 == 0 \land m + n == 0 \land m < -\frac{1}{2}$, then

$$\int (a + b\,\mathrm{Sec}[e + f\,x])^m\,(d\,\mathrm{Sec}[e + f\,x])^n\,dx \longrightarrow$$
$$-\frac{b\,d\,\mathrm{Tan}[e + f\,x]\,(a + b\,\mathrm{Sec}[e + f\,x])^m\,(d\,\mathrm{Sec}[e + f\,x])^{n-1}}{a\,f\,(2\,m + 1)} + \frac{d\,(m + 1)}{b\,(2\,m + 1)}\int (a + b\,\mathrm{Sec}[e + f\,x])^{m+1}\,(d\,\mathrm{Sec}[e + f\,x])^{n-1}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  b*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)/(a*f*(2*m+1)) +
  d*(m+1)/(b*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && EqQ[m+n,0] && LtQ[m,-1/2] && IntegerQ[2*m]
```

7. $\int (a + b\,\text{Sec}[e + f\,x])^m\,(d\,\text{Sec}[e + f\,x])^n\,dx$ when $a^2 - b^2 == 0 \land m + n + 1 == 0$

1: $\int (a + b\,\text{Sec}[e + f\,x])^m\,(d\,\text{Sec}[e + f\,x])^n\,dx$ when $a^2 - b^2 == 0 \land m + n + 1 == 0 \land m < -\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 2b with $A \to 1,\ B \to 0,\ n \to -m - 2,\ p \to 0$

Rule: If $a^2 - b^2 == 0 \land m + n + 1 == 0 \land m < -\frac{1}{2}$, then

$$\int (a + b\,\text{Sec}[e + f\,x])^m\,(d\,\text{Sec}[e + f\,x])^n\,dx \longrightarrow$$

$$\frac{\text{Tan}[e + f\,x]\,(a + b\,\text{Sec}[e + f\,x])^m\,(d\,\text{Sec}[e + f\,x])^n}{f\,(2\,m + 1)} + \frac{m}{a\,(2\,m + 1)} \int (a + b\,\text{Sec}[e + f\,x])^{m+1}\,(d\,\text{Sec}[e + f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*(2*m+1)) +
  m/(a*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && EqQ[m+n+1,0] && LtQ[m,-1/2]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2=0\,\land\,m+n+1=0\,\land\,m\nless-\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 1c with $A\to 1,\ B\to 0,\ m\to -n-2,\ p\to 0$

Rule: If $a^2-b^2=0\,\land\,m+n+1=0\,\land\,m\nless-\frac{1}{2}$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\frac{\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n}{f\,(m+1)}+\frac{a\,m}{b\,d\,(m+1)}\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n+1}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*(m+1)) +
  a*m/(b*d*(m+1))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && EqQ[m+n+1,0] && Not[LtQ[m,-1/2]]
```

8. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\land\,m>1$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\land\,m>1\,\land\,n<-1$

Derivation: Singly degenerate secant recurrence 1a with $A\to a,\ B\to b,\ m\to m-1,\ p\to 0$

Rule: If $a^2-b^2==0\,\land\,m>1\,\land\,n<-1$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$
$$-\frac{b^2\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-2}\,(d\,\text{Sec}[e+f\,x])^n}{f\,n}-$$
$$\frac{a}{d\,n}\int (a+b\,\text{Sec}[e+f\,x])^{m-2}\,(d\,\text{Sec}[e+f\,x])^{n+1}\,(b\,(m-2\,n-2)-a\,(m+2\,n-1)\,\text{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  b^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^n/(f*n) -
  a/(d*n)*Int[(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^(n+1)*(b*(m-2*n-2)-a*(m+2*n-1)*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && GtQ[m,1] && (LtQ[n,-1] || EqQ[m,3/2] && EqQ[n,-1/2]) && IntegerQ[2*m]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\wedge\,m>1\,\wedge\,n\not<-1\,\wedge\,m+n-1\neq 0$

Derivation: Singly degenerate secant recurrence 1b with $A\to a,\ B\to b,\ m\to m-1,\ p\to 0$

Rule: If $a^2-b^2==0\,\wedge\,m>1\,\wedge\,n\not<-1\,\wedge\,m+n-1\neq 0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\frac{b^2\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-2}\,(d\,\text{Sec}[e+f\,x])^n}{f\,(m+n-1)}+$$
$$\frac{b}{m+n-1}\int (a+b\,\text{Sec}[e+f\,x])^{m-2}\,(d\,\text{Sec}[e+f\,x])^n\,(b\,(m+2\,n-1)+a\,(3\,m+2\,n-4)\,\text{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -b^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^n/(f*(m+n-1)) +
  b/(m+n-1)*Int[(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^n*(b*(m+2*n-1)+a*(3*m+2*n-4)*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,n},x] && EqQ[a^2-b^2,0] && GtQ[m,1] && NeQ[m+n-1,0] && IntegerQ[2*m]
```

9. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\wedge\,m<-1$

1. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\wedge\,m<-1\,\wedge\,n>1$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\wedge\,m<-1\,\wedge\,1<n<2$

Derivation: Singly degenerate secant recurrence 2a with $A\to 1,\ B\to 0,\ p\to 0$

Rule: If $a^2-b^2==0\,\wedge\,m<-1\,\wedge\,1<n<2$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$
$$-\frac{b\,d\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n-1}}{a\,f\,(2\,m+1)}-$$

$$\frac{d}{a\,b\,(2\,m+1)}\int(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-1}\,(a\,(n-1)-b\,(m+n)\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  b*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)/(a*f*(2*m+1)) -
  d/(a*b*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)*(a*(n-1)-b*(m+n)*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && LtQ[m,-1] && LtQ[1,n,2] && (IntegersQ[2*m,2*n] || IntegerQ[m])
```

2: $\int(a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\wedge m<-1\wedge n>2$

Derivation: Singly degenerate secant recurrence 2a with $A\to c,\ B\to d,\ n\to n-1,\ p\to 0$

Rule: If $a^2-b^2==0\wedge m<-1\wedge n>2$, then

$$\int(a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx\longrightarrow$$

$$\frac{d^2\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^{n-2}}{f\,(2\,m+1)}+$$

$$\frac{d^2}{a\,b\,(2\,m+1)}\int(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-2}\,(b\,(n-2)+a\,(m-n+2)\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -d^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-2)/(f*(2*m+1)) +
  d^2/(a*b*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-2)*(b*(n-2)+a*(m-n+2)*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && LtQ[m,-1] && GtQ[n,2] && (IntegersQ[2*m,2*n] || IntegerQ[m])
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\wedge\,m<-1\,\wedge\,n\not>0$

Derivation: Singly degenerate secant recurrence 2b with $A\to 1,\ B\to 0,\ p\to 0$

Rule: If $a^2-b^2==0\,\wedge\,m<-1\,\wedge\,n\not>0$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\frac{\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n}{f\,(2\,m+1)}+$$
$$\frac{1}{a^2\,(2\,m+1)}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^n\,(a\,(2\,m+n+1)-b\,(m+n+1)\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*(2*m+1)) +
  1/(a^2*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n*(a*(2*m+n+1)-b*(m+n+1)*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,n},x] && EqQ[a^2-b^2,0] && LtQ[m,-1] && (IntegersQ[2*m,2*n] || IntegerQ[m])
```

10. $\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx$ when $a^2-b^2=0$

1: $\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx$ when $a^2-b^2=0\land n>1$

Derivation: Singly degenerate secant recurrence 2a with $A\to c,\ B\to d,\ m\to -1,\ n\to n-1,\ p\to 0$

Rule: If $a^2-b^2=0\land n>1$, then

$$\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx \longrightarrow -\frac{d^2\,\mathrm{Tan}[e+f\,x]\,(d\,\mathrm{Sec}[e+f\,x])^{n-2}}{f\,(a+b\,\mathrm{Sec}[e+f\,x])}-\frac{d^2}{a\,b}\int (d\,\mathrm{Sec}[e+f\,x])^{n-2}\,(b\,(n-2)-a\,(n-1)\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  d^2*Cot[e+f*x]*(d*Csc[e+f*x])^(n-2)/(f*(a+b*Csc[e+f*x])) -
  d^2/(a*b)*Int[(d*Csc[e+f*x])^(n-2)*(b*(n-2)-a*(n-1)*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && GtQ[n,1]
```

2: $\int \frac{(d \operatorname{Sec}[e+f x])^n}{a+b \operatorname{Sec}[e+f x]} \, dx$ when $a^2-b^2==0 \land n<0$

Derivation: Singly degenerate secant recurrence 2b with $A \to 1$, $B \to 0$, $m \to -1$, $p \to 0$

Rule: If $a^2-b^2==0 \land n<0$, then

$$\int \frac{(d \operatorname{Sec}[e+f x])^n}{a+b \operatorname{Sec}[e+f x]} \, dx \longrightarrow -\frac{\operatorname{Tan}[e+f x] \, (d \operatorname{Sec}[e+f x])^n}{f \, (a+b \operatorname{Sec}[e+f x])} - \frac{1}{a^2} \int (d \operatorname{Sec}[e+f x])^n \, (a \, (n-1) - b \, n \operatorname{Sec}[e+f x]) \, dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*(a+b*Csc[e+f*x])) -
  1/a^2*Int[(d*Csc[e+f*x])^n*(a*(n-1)-b*n*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && LtQ[n,0]
```

3: $\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx$ when $a^2-b^2==0$

Derivation: Singly degenerate secant recurrence 2a with $A\to 1$, $B\to 0$, $m\to -1$, $p\to 0$

Rule: If $a^2-b^2==0$, then

$$\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx \longrightarrow \frac{b\,d\,\mathrm{Tan}[e+f\,x]\,(d\,\mathrm{Sec}[e+f\,x])^{n-1}}{a\,f\,(a+b\,\mathrm{Sec}[e+f\,x])}+\frac{d\,(n-1)}{a\,b}\int (d\,\mathrm{Sec}[e+f\,x])^{n-1}\,(a-b\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -b*d*Cot[e+f*x]*(d*Csc[e+f*x])^(n-1)/(a*f*(a+b*Csc[e+f*x])) +
  d*(n-1)/(a*b)*Int[(d*Csc[e+f*x])^(n-1)*(a-b*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,n},x] && EqQ[a^2-b^2,0]
```

11. $\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2==0$

1. $\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2==0\wedge n>1$

1: $\int \frac{(d\,\mathrm{Sec}[e+f\,x])^{3/2}}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2==0$

Derivation: Algebraic expansion

Basis: $\frac{d\,z}{\sqrt{a+b\,z}}==\frac{d\,\sqrt{a+b\,z}}{b}-\frac{a\,d}{b\,\sqrt{a+b\,z}}$

Rule: If $a^2-b^2==0$, then

$$\int \frac{(d\,\text{Sec}[e+f\,x])^{3/2}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx \longrightarrow \frac{d}{b}\int\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}\,dx-\frac{a\,d}{b}\int\frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^(3/2)/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  d/b*Int[Sqrt[a+b*Csc[e+f*x]]*Sqrt[d*Csc[e+f*x]],x] -
  a*d/b*Int[Sqrt[d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0]
```

2: $\int \frac{(d\,\text{Sec}[e+f\,x])^{n}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2==0\land n>2$

Derivation: Singly degenerate secant recurrence 2c with $A\to c,\ B\to d,\ m\to\frac{1}{2},\ n\to n-1,\ p\to 0$

Rule: If $a^2-b^2==0\land n>2$, then

$$\int \frac{(d\,\text{Sec}[e+f\,x])^{n}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx \longrightarrow$$

$$\frac{2\,d^2\,\text{Tan}[e+f\,x]\,(d\,\text{Sec}[e+f\,x])^{n-2}}{f\,(2\,n-3)\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}+\frac{d^2}{b\,(2\,n-3)}\int\frac{(d\,\text{Sec}[e+f\,x])^{n-2}\,(2\,b\,(n-2)-a\,\text{Sec}[e+f\,x])}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*d^2*Cot[e+f*x]*(d*Csc[e+f*x])^(n-2)/(f*(2*n-3)*Sqrt[a+b*Csc[e+f*x]]) +
  d^2/(b*(2*n-3))*Int[(d*Csc[e+f*x])^(n-2)*(2*b*(n-2)-a*Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && GtQ[n,2] && IntegerQ[2*n]
```

2: $\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2==0\land n<0$

Derivation: Singly degenerate secant recurrence 1c with $A\to 1$, $B\to 0$, $p\to 0$

Rule: If $a^2-b^2==0\land n<0$, then

$$\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx \to -\frac{\mathrm{Tan}[e+f\,x]\,(d\,\mathrm{Sec}[e+f\,x])^n}{f\,n\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}+\frac{1}{2\,b\,d\,n}\int \frac{(d\,\mathrm{Sec}[e+f\,x])^{n+1}\,(a+b\,(2\,n+1)\,\mathrm{Sec}[e+f\,x])}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*n*Sqrt[a+b*Csc[e+f*x]]) +
  1/(2*b*d*n)*Int[(d*Csc[e+f*x])^(n+1)*(a+b*(2*n+1)*Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && LtQ[n,0] && IntegerQ[2*n]
```

12: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\land n>2\land m+n-1\neq 0$

Derivation: Singly degenerate secant recurrence 2c with $A\to c,\ B\to d,\ n\to n-1,\ p\to 0$

Rule: If $a^2-b^2==0\land n>2\land m+n-1\neq 0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\longrightarrow$$

$$\frac{d^2\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n-2}}{f\,(m+n-1)}+\frac{d^2}{b\,(m+n-1)}\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n-2}\,(b\,(n-2)+a\,m\,\text{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -d^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-2)/(f*(m+n-1)) +
  d^2/(b*(m+n-1))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-2)*(b*(n-2)+a*m*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,m},x] && EqQ[a^2-b^2,0] && GtQ[n,2] && NeQ[m+n-1,0] && IntegerQ[n]
```

13. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\land m\notin\mathbb{Z}\land a>0$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\land m\notin\mathbb{Z}\land a>0\land n\notin\mathbb{Z}\land\frac{a\,d}{b}>0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2==0$, then $\partial_x\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}==0$

Basis: If $a^2-b^2==0$, then $-\frac{a^2\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}\;\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}==1$

Basis: If $a>0$, then $\frac{\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-\frac{1}{2}}\,\left(\frac{b}{a}\,\text{Sec}[e+f\,x]\right)^n}{\sqrt{a-b\,\text{Sec}[e+f\,x]}}==$

$$-\frac{1}{a^n\,f}\,\text{Subst}\left[\frac{(a-x)^{n-1}\,(2\,a-x)^{m-\frac{1}{2}}}{\sqrt{x}},\,x,\,a-b\,\text{Sec}[e+f\,x]\right]\,\partial_x\,(a-b\,\text{Sec}[e+f\,x])$$

Rule: If $a^2-b^2==0\land m\notin\mathbb{Z}\land a>0\land n\notin\mathbb{Z}\land\frac{a\,d}{b}>0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\;\longrightarrow$$

$$-\frac{a^2\,\left(\frac{a\,d}{b}\right)^n\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}\int\frac{\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-\frac{1}{2}}\,\left(\frac{b}{a}\,\text{Sec}[e+f\,x]\right)^n}{\sqrt{a-b\,\text{Sec}[e+f\,x]}}\,dx\;\longrightarrow$$

$$\frac{\left(\frac{a\,d}{b}\right)^n\,\text{Tan}[e+f\,x]}{a^{n-2}\,f\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}\,\text{Subst}\left[\int\frac{(a-x)^{n-1}\,(2\,a-x)^{m-\frac{1}{2}}}{\sqrt{x}}\,dx,\,x,\,a-b\,\text{Sec}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -(a*d/b)^n*Cot[e+f*x]/(a^(n-2)*f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[a-b*Csc[e+f*x]])*
   Subst[Int[(a-x)^(n-1)*(2*a-x)^(m-1/2)/Sqrt[x],x],x,a-b*Csc[e+f*x]] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && GtQ[a,0] && Not[IntegerQ[n]] && GtQ[a*d/b,0]
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\ \land\ m\notin\mathbb{Z}\ \land\ a>0\ \land\ n\notin\mathbb{Z}\ \land\ \frac{a\,d}{b}<0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2==0$, then $\partial_x \frac{\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}==0$

Basis: If $a^2-b^2==0$, then $-\frac{a^2\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}\;\frac{\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}==1$

Basis: If $a>0$, then $\frac{\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m-\frac{1}{2}}\,\left(-\frac{b}{a}\,\mathrm{Sec}[e+f\,x]\right)^n}{\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}==$

$$-\frac{1}{a^n\,f}\,\mathrm{Subst}\left[\frac{x^{m-\frac{1}{2}}\,(a-x)^{n-1}}{\sqrt{2\,a-x}},\,x,\,a+b\,\mathrm{Sec}[e+f\,x]\right]\,\partial_x\,(a+b\,\mathrm{Sec}[e+f\,x])$$

Rule: If $a^2-b^2==0\ \land\ m\notin\mathbb{Z}\ \land\ a>0\ \land\ n\notin\mathbb{Z}\ \land\ \frac{a\,d}{b}<0$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx\ \longrightarrow$$

$$-\frac{a^2\,\left(-\frac{a\,d}{b}\right)^n\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}\int\frac{\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m-\frac{1}{2}}\,\left(-\frac{b}{a}\,\mathrm{Sec}[e+f\,x]\right)^n}{\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}\,dx\ \longrightarrow$$

$$\frac{\left(-\frac{a\,d}{b}\right)^n\,\mathrm{Tan}[e+f\,x]}{a^{n-1}\,f\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}\,\mathrm{Subst}\left[\int\frac{x^{m-\frac{1}{2}}\,(a-x)^{n-1}}{\sqrt{2\,a-x}}\,dx,\,x,\,a+b\,\mathrm{Sec}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -(-a*d/b)^n*Cot[e+f*x]/(a^(n-1)*f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[a-b*Csc[e+f*x]])*
    Subst[Int[x^(m-1/2)*(a-x)^(n-1)/Sqrt[2*a-x],x],x,a+b*Csc[e+f*x]] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && GtQ[a,0] && Not[IntegerQ[n]] && LtQ[a*d/b,0]
```

3: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2=0\,\land\,m\notin\mathbb{Z}\,\land\,a>0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2=0$, then $\partial_x \frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}=0$

Basis: If $a^2-b^2=0$, then $-\frac{a^2\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}\;\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}=1$

Basis: $\text{Tan}[e+f\,x]\,F[\text{Sec}[e+f\,x]]=\frac{1}{f}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,\text{Sec}[e+f\,x]\right]\,\partial_x\text{Sec}[e+f\,x]$

Note: If $a>0$, then $\frac{(d\,x)^{n-1}\,(a+b\,x)^{m-\frac{1}{2}}}{\sqrt{a-b\,x}}$ is integrable without the need for additional piecewise constant factors.

Rule: If $a^2-b^2=0\,\land\,m\notin\mathbb{Z}\,\land\,a>0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\,\to$$

$$-\frac{a^2\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}\int\frac{\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-\frac{1}{2}}\,(d\,\text{Sec}[e+f\,x])^n}{\sqrt{a-b\,\text{Sec}[e+f\,x]}}\,dx\,\to$$

$$-\frac{a^2\,d\,\text{Tan}[e+f\,x]}{f\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}\,\text{Subst}\left[\int\frac{(d\,x)^{n-1}\,(a+b\,x)^{m-\frac{1}{2}}}{\sqrt{a-b\,x}}\,dx,\,x,\,\text{Sec}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^2*d*Cot[e+f*x]/(f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[a-b*Csc[e+f*x]])*
    Subst[Int[(d*x)^(n-1)*(a+b*x)^(m-1/2)/Sqrt[a-b*x],x],x,Csc[e+f*x]] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && GtQ[a,0]
```

14: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2==0\land m\notin\mathbb{Z}\land a\not>0$

Derivation: Piecewise constant extraction

Basis: If $\partial_x \frac{(a+b\,\text{Sec}[e+f\,x])^m}{\left(1+\frac{b}{a}\,\text{Sec}[e+f\,x]\right)^m}==0$

Rule: If $a^2-b^2==0\land m\notin\mathbb{Z}\land n\notin\mathbb{Z}\land \frac{a\,d}{b}>0\land a\not>0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\to\frac{a^{\text{IntPart}[m]}\,(a+b\,\text{Sec}[e+f\,x])^{\text{FracPart}[m]}}{\left(1+\frac{b}{a}\,\text{Sec}[e+f\,x]\right)^{\text{FracPart}[m]}}\int\left(1+\frac{b}{a}\,\text{Sec}[e+f\,x]\right)^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^IntPart[m]*(a+b*Csc[e+f*x])^FracPart[m]/(1+b/a*Csc[e+f*x])^FracPart[m]*Int[(1+b/a*Csc[e+f*x])^m*(d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && Not[GtQ[a,0]]
```

6. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0$

1. $\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0$

1. $\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\land m>0$

1: $\int \text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\sqrt{a+b\,z}==\frac{a-b}{\sqrt{a+b\,z}}+\frac{b\,(1+z)}{\sqrt{a+b\,z}}$

Rule: If $a^2-b^2\neq 0$, then

$$\int \operatorname{Sec}[e+f x] \sqrt{a+b \operatorname{Sec}[e+f x]}\, dx \longrightarrow (a-b) \int \frac{\operatorname{Sec}[e+f x]}{\sqrt{a+b \operatorname{Sec}[e+f x]}}\, dx + b \int \frac{\operatorname{Sec}[e+f x]\, (1+\operatorname{Sec}[e+f x])}{\sqrt{a+b \operatorname{Sec}[e+f x]}}\, dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  (a-b)*Int[Csc[e+f*x]/Sqrt[a+b*Csc[e+f*x]],x] + b*Int[Csc[e+f*x]*(1+Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \operatorname{Sec}[e+f x]\, (a+b \operatorname{Sec}[e+f x])^m\, dx$ when $a^2-b^2 \neq 0 \land m>1$

Derivation: Cosecant recurrence 1b with $c \to a\, c,\ d \to b\, c + a\, d,\ C \to b\, d,\ m \to 0,\ n \to n-1$

Rule: If $a^2-b^2 \neq 0 \land m>1$, then

$$\int \operatorname{Sec}[e+f x]\, (a+b \operatorname{Sec}[e+f x])^m\, dx \longrightarrow$$
$$\frac{b \operatorname{Tan}[e+f x]\, (a+b \operatorname{Sec}[e+f x])^{m-1}}{f\, m} + \frac{1}{m} \int \operatorname{Sec}[e+f x]\, (a+b \operatorname{Sec}[e+f x])^{m-2} \left(b^2 (m-1) + a^2 m + a\, b\, (2 m-1) \operatorname{Sec}[e+f x]\right) dx$$

–

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -b*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)/(f*m) +
  1/m*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m-2)*(b^2*(m-1)+a^2*m+a*b*(2*m-1)*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0] && GtQ[m,1] && IntegerQ[2*m]
```

2. $\int \operatorname{Sec}[e+f x]\, (a+b \operatorname{Sec}[e+f x])^m\, dx$ when $a^2-b^2 \neq 0 \land m<0$

1. $\int \frac{\operatorname{Sec}[e+f x]}{a+b \operatorname{Sec}[e+f x]}\, dx$ when $a^2-b^2 \neq 0$

x: $\int \frac{\text{Sec}[e+f x]}{a+b\,\text{Sec}[e+f x]}\,dx$ when $a^2-b^2\neq 0$

Derivation: Integration by substitution

Basis: $\frac{\text{Sec}[e+f x]}{a+b\,\text{Sec}[e+f x]} == \frac{2}{f}\,\text{Subst}\left[\frac{1}{a+b-(a-b)\,x^2},\,x,\,\frac{\text{Tan}[e+f x]}{1+\text{Sec}[e+f x]}\right]\partial_x\frac{\text{Tan}[e+f x]}{1+\text{Sec}[e+f x]}$

Rule: This rule may be preferable to the following one, but will require numerous changes to the test suite.

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{\text{Sec}[e+f x]}{a+b\,\text{Sec}[e+f x]}\,dx \longrightarrow \frac{2}{f}\,\text{Subst}\left[\int \frac{1}{a+b-(a-b)\,x^2}\,dx,\,x,\,\frac{\text{Tan}[e+f x]}{1+\text{Sec}[e+f x]}\right]$$

Program code:

```
(* Int[csc[e_.+f_.*x_]/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -2/f*Subst[Int[1/(a+b-(a-b)*x^2),x],x,Cot[e+f*x]/(1+Csc[e+f*x])] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0] *)
```

1: $\int \frac{\text{Sec}[e+f x]}{a+b\,\text{Sec}[e+f x]}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic simplification

Basis: $\frac{z}{a+b z} == \frac{1}{b\left(1+\frac{a}{b} z^{-1}\right)}$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{\text{Sec}[e+f x]}{a+b\,\text{Sec}[e+f x]}\,dx \to \frac{1}{b}\int \frac{1}{1+\frac{a}{b}\,\text{Cos}[e+f x]}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  1/b*Int[1/(1+a/b*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x\left(\frac{1}{\text{Tan}[e+f x]}\sqrt{\frac{b\,(1-\text{Sec}[e+f x])}{a+b}}\sqrt{-\frac{b\,(1+\text{Sec}[e+f x])}{a-b}}\right) == 0$

Basis: $\text{Sec}[e+f x]\,\text{Tan}[e+f x]\,F[\text{Sec}[e+f x]] == \frac{1}{f}\,\text{Subst}[F[x],\,x,\,\text{Sec}[e+f x]]\,\partial_x\text{Sec}[e+f x]$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx \to \frac{1}{\text{Tan}[e+f x]}\sqrt{\frac{b\,(1-\text{Sec}[e+f x])}{a+b}}\sqrt{-\frac{b\,(1+\text{Sec}[e+f x])}{a-b}}\int \frac{\text{Sec}[e+f x]\,\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\sqrt{\frac{b}{a+b}-\frac{b\,\text{Sec}[e+f x]}{a+b}}\sqrt{-\frac{b}{a-b}-\frac{b\,\text{Sec}[e+f x]}{a-b}}}\,dx$$

$$\to \frac{1}{f\,\text{Tan}[e+f x]}\sqrt{\frac{b\,(1-\text{Sec}[e+f x])}{a+b}}\sqrt{-\frac{b\,(1+\text{Sec}[e+f x])}{a-b}}\;\text{Subst}\Big[\int\frac{1}{\sqrt{a+b x}\,\sqrt{\frac{b}{a+b}-\frac{b x}{a+b}}\,\sqrt{-\frac{b}{a-b}-\frac{b x}{a-b}}}\,dx,\,x,\,\text{Sec}[e+f x]\Big]$$

$$\to \frac{2\sqrt{a+b}}{b\,f\,\text{Tan}[e+f x]}\sqrt{\frac{b\,(1-\text{Sec}[e+f x])}{a+b}}\sqrt{-\frac{b\,(1+\text{Sec}[e+f x])}{a-b}}\;\text{EllipticF}\Big[\text{ArcSin}\Big[\frac{\sqrt{a+b\,\text{Sec}[e+f x]}}{\sqrt{a+b}}\Big],\,\frac{a+b}{a-b}\Big]$$

Program code:

```
Int[csc[e_.+f_.*x_]/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*Rt[a+b,2]/(b*f*Cot[e+f*x])*Sqrt[(b*(1-Csc[e+f*x]))/(a+b)]*Sqrt[-b*(1+Csc[e+f*x])/(a-b)]*
    EllipticF[ArcSin[Sqrt[a+b*Csc[e+f*x]]/Rt[a+b,2]],(a+b)/(a-b)] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0]
```

3: $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,dx$ when $a^2-b^2\neq 0 \land m<-1$

Derivation: Cosecant recurrence 2b with $C\to 0,\ m\to 0$

Rule: If $a^2-b^2\neq 0 \land m<-1$, then

$$\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,dx \longrightarrow$$

$$\frac{b\,\text{Tan}[e+f x]\,(a+b\,\text{Sec}[e+f x])^{m+1}}{f\,(m+1)\,(a^2-b^2)}+\frac{1}{(m+1)\,(a^2-b^2)}\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^{m+1}\,(a\,(m+1)-b\,(m+2)\,\text{Sec}[e+f x])\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -b*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(f*(m+1)*(a^2-b^2)) +
  1/((m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(a*(m+1)-b*(m+2)*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && IntegerQ[2*m]
```

3: $\int \operatorname{Sec}[e+f x]\,(a+b\operatorname{Sec}[e+f x])^m\,dx$ when $a^2-b^2\neq 0\,\wedge\,2m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\operatorname{Tan}[e+f x]}{\sqrt{1+\operatorname{Sec}[e+f x]}\,\sqrt{1-\operatorname{Sec}[e+f x]}} == 0$

Basis: $-\frac{\operatorname{Tan}[e+f x]}{\sqrt{1+\operatorname{Sec}[e+f x]}\,\sqrt{1-\operatorname{Sec}[e+f x]}}\,\frac{\operatorname{Tan}[e+f x]}{\sqrt{1+\operatorname{Sec}[e+f x]}\,\sqrt{1-\operatorname{Sec}[e+f x]}} == 1$

Basis: $\operatorname{Tan}[e+f x]\,F[\operatorname{Sec}[e+f x]] == \frac{1}{f}\operatorname{Subst}\left[\frac{F[x]}{x},\,x,\,\operatorname{Sec}[e+f x]\right]\partial_x\operatorname{Sec}[e+f x]$

Rule: If $a^2-b^2\neq 0\,\wedge\,2m\notin\mathbb{Z}$, then

$$\int \operatorname{Sec}[e+f x]\,(a+b\operatorname{Sec}[e+f x])^m\,dx \rightarrow -\frac{\operatorname{Tan}[e+f x]}{\sqrt{1+\operatorname{Sec}[e+f x]}\,\sqrt{1-\operatorname{Sec}[e+f x]}}\int\frac{\operatorname{Tan}[e+f x]\,\operatorname{Sec}[e+f x]\,(a+b\operatorname{Sec}[e+f x])^m}{\sqrt{1+\operatorname{Sec}[e+f x]}\,\sqrt{1-\operatorname{Sec}[e+f x]}}\,dx$$

$$\rightarrow -\frac{\operatorname{Tan}[e+f x]}{f\sqrt{1+\operatorname{Sec}[e+f x]}\,\sqrt{1-\operatorname{Sec}[e+f x]}}\operatorname{Subst}\left[\int\frac{(a+b x)^m}{\sqrt{1+x}\,\sqrt{1-x}}\,dx,\,x,\,\operatorname{Sec}[e+f x]\right]$$

–

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
 Cot[e+f*x]/(f*Sqrt[1+Csc[e+f*x]]*Sqrt[1-Csc[e+f*x]])*Subst[Int[(a+b*x)^m/(Sqrt[1+x]*Sqrt[1-x]),x],x,Csc[e+f*x]] /;
FreeQ[{a,b,e,f,m},x] && NeQ[a^2-b^2,0] && Not[IntegerQ[2*m]]
```

2. $\int \text{Sec}[e+f x]^2 (a+b \text{Sec}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0$

1: $\int \text{Sec}[e+f x]^2 (a+b \text{Sec}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \wedge m>0$

Reference: G&R 2.551.1 inverted

Derivation: Nondegenerate secant recurrence 1b with $A \to a c$, $B \to b c + a d$, $C \to b d$, $m \to 0$, $n \to n-1$, $p \to 0$

Rule: If $a^2-b^2 \neq 0 \wedge m>0$, then

$$\int \text{Sec}[e+f x]^2 (a+b \text{Sec}[e+f x])^m \, dx \to \frac{\text{Tan}[e+f x] (a+b \text{Sec}[e+f x])^m}{f (m+1)} + \frac{m}{m+1} \int \text{Sec}[e+f x] (a+b \text{Sec}[e+f x])^{m-1} (b+a \text{Sec}[e+f x]) \, dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) +
  m/(m+1)*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(b+a*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0] && GtQ[m,0]
```

2: $\int \mathrm{Sec}[e+f\,x]^2\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\,\wedge\,m<-1$

Reference: G&R 2.551.1

Derivation: Nondegenerate secant recurrence 1a with $A\to c,\ B\to d,\ C\to 0,\ n\to 0,\ p\to 0$

Rule: If $a^2-b^2\neq 0\,\wedge\,m<-1$, then

$$\int \mathrm{Sec}[e+f\,x]^2\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,dx\,\longrightarrow$$

$$-\frac{a\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}}{f\,(m+1)\,(a^2-b^2)}-\frac{1}{(m+1)\,(a^2-b^2)}\int \mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(b\,(m+1)-a\,(m+2)\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  a*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(f*(m+1)*(a^2-b^2)) -
  1/((m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(b*(m+1)-a*(m+2)*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

3: $\int \frac{\text{Sec}[e+f x]^2}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{\text{Sec}[e+f x]^2}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx \to -\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx+\int \frac{\text{Sec}[e+f x]\,(1+\text{Sec}[e+f x])}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -Int[Csc[e+f*x]/Sqrt[a+b*Csc[e+f*x]],x] +
  Int[Csc[e+f*x]*(1+Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0]
```

4: $\int \text{Sec}[e+f\,x]^2\,(a+b\,\text{Sec}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $z^2 == -\frac{a\,z}{b}+\frac{1}{b}\,z\,(a+b\,z)$

Rule: If $a^2-b^2\neq 0$, then

$$\int \text{Sec}[e+f\,x]^2\,(a+b\,\text{Sec}[e+f\,x])^m\,dx \longrightarrow -\frac{a}{b}\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,dx+\frac{1}{b}\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -a/b*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m,x] + 1/b*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f,m},x] && NeQ[a^2-b^2,0]
```

3. $\int \operatorname{Sec}[e+f x]^3 (a+b \operatorname{Sec}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0$

1: $\int \operatorname{Sec}[e+f x]^3 (a+b \operatorname{Sec}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land m<-1$

Derivation: Nondegenerate secant recurrence 1a with $A \to c^2$, $B \to 2 c d$, $C \to d^2$, $n \to 0$, $p \to 0$

Rule: If $a^2-b^2 \neq 0 \land m<-1$, then

$$\int \operatorname{Sec}[e+f x]^3 (a+b \operatorname{Sec}[e+f x])^m \, dx \longrightarrow$$
$$\frac{a^2 \operatorname{Tan}[e+f x] (a+b \operatorname{Sec}[e+f x])^{m+1}}{b f (m+1) (a^2-b^2)}+$$
$$\frac{1}{b (m+1) (a^2-b^2)} \int \operatorname{Sec}[e+f x] (a+b \operatorname{Sec}[e+f x])^{m+1} \left(a b (m+1)-\left(a^2+b^2 (m+1)\right) \operatorname{Sec}[e+f x]\right) dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^3*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -a^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+1)*(a^2-b^2)) +
  1/(b*(m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*Simp[a*b*(m+1)-(a^2+b^2*(m+1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

2: $\int \text{Sec}[e+f x]^3 (a+b\,\text{Sec}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \wedge m \nless -1$

Derivation: Nondegenerate secant recurrence 1b with $A \to a^2$, $B \to 2 a b$, $C \to b^2$, $m \to 0$, $p \to 0$

Rule: If $a^2-b^2 \neq 0 \wedge m \nless -1$, then

$$\int \text{Sec}[e+f x]^3 (a+b\,\text{Sec}[e+f x])^m \, dx \longrightarrow$$

$$\frac{\text{Tan}[e+f x] (a+b\,\text{Sec}[e+f x])^{m+1}}{b f (m+2)}+\frac{1}{b (m+2)} \int \text{Sec}[e+f x] (a+b\,\text{Sec}[e+f x])^m (b (m+1)-a\,\text{Sec}[e+f x]) \, dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^3*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m*(b*(m+1)-a*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f,m},x] && NeQ[a^2-b^2,0] && Not[LtQ[m,-1]]
```

4. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\wedge\,m>2$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\wedge\,m>2\,\wedge\,n<-1$

Derivation: Nondegenerate secant recurrence 1a with $A\to c^2,\ B\to 2\,c\,d,\ C\to d^2,\ n\to n-2,\ p\to 0$

Rule: If $a^2-b^2\neq 0\,\wedge\,m>2\,\wedge\,n<-1$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$
$$-\frac{a^2\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-2}\,(d\,\text{Sec}[e+f\,x])^n}{f\,n}-$$
$$\frac{1}{d\,n}\int (a+b\,\text{Sec}[e+f\,x])^{m-3}\,(d\,\text{Sec}[e+f\,x])^{n+1}\,\left(a^2\,b\,(m-2\,n-2)-a\,\left(3\,b^2\,n+a^2\,(n+1)\right)\,\text{Sec}[e+f\,x]-b\,\left(b^2\,n+a^2\,(m+n-1)\right)\,\text{Sec}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  a^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^n/(f*n) -
  1/(d*n)*Int[(a+b*Csc[e+f*x])^(m-3)*(d*Csc[e+f*x])^(n+1)*
    Simp[a^2*b*(m-2*n-2)-a*(3*b^2*n+a^2*(n+1))*Csc[e+f*x]-b*(b^2*n+a^2*(m+n-1))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && GtQ[m,2] && (IntegerQ[m] && LtQ[n,-1] || IntegersQ[m+1/2,2*n] && LeQ[n,-1])
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\wedge\,m>2\,\wedge\,n\not<-1$

Derivation: Nondegenerate secant recurrence 1b with $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to m-2,\ p\to 0$

Rule: If $a^2-b^2\neq 0\,\wedge\,m>2\,\wedge\,n\not<-1$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\frac{b^2\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-2}\,(d\,\text{Sec}[e+f\,x])^n}{f\,(m+n-1)}+$$
$$\frac{1}{m+n-1}\int (a+b\,\text{Sec}[e+f\,x])^{m-3}\,(d\,\text{Sec}[e+f\,x])^n\,\cdot$$

$$\left(a^3 (m+n-1) + a b^2 n + b \left(b^2 (m+n-2) + 3 a^2 (m+n-1)\right) \text{Sec}[e+f x] + a b^2 (3 m+2 n-4) \text{Sec}[e+f x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -b^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^n/(f*(m+n-1)) +
  1/(d*(m+n-1))*Int[(a+b*Csc[e+f*x])^(m-3)*(d*Csc[e+f*x])^n*
    Simp[a^3*d*(m+n-1)+a*b^2*d*n+b*(b^2*d*(m+n-2)+3*a^2*d*(m+n-1))*Csc[e+f*x]+a*b^2*d*(3*m+2*n-4)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,n},x] && NeQ[a^2-b^2,0] && GtQ[m,2] && (IntegerQ[m] || IntegersQ[2*m,2*n]) && Not[IGtQ[n,2] && Not[IntegerQ[m]]]
```

5. $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\wedge\,m<-1$

1. $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,n>0$

1: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,0<n<1$

Derivation: Nondegenerate secant recurrence 1a with $A\to 1,\ B\to 0,\ C\to 0,\ p\to 0$

Derivation: Nondegenerate secant recurrence 1c with $A\to c,\ B\to d,\ C\to 0,\ n\to n-1,\ p\to 0$

Rule: If $a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,0<n<1$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\frac{b\,d\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-1}}{f\,(m+1)\,(a^2-b^2)}+$$

$$\frac{1}{(m+1)\,(a^2-b^2)}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-1}\,\left(b\,d\,(n-1)+a\,d\,(m+1)\,\mathrm{Sec}[e+f\,x]-b\,d\,(m+n+1)\,\mathrm{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -b*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)/(f*(m+1)*(a^2-b^2)) +
  1/((m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)*
    Simp[b*d*(n-1)+a*d*(m+1)*Csc[e+f*x]-b*d*(m+n+1)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && LtQ[0,n,1] && IntegersQ[2*m,2*n]
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,1<n<2$

Derivation: Nondegenerate secant recurrence 1a with $A\to c,\ B\to d,\ C\to 0,\ n\to n-1,\ p\to 0$

Rule: If $a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,1<n<2$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx \longrightarrow$$
$$-\frac{a\,d^2\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-2}}{f\,(m+1)\,(a^2-b^2)}-$$
$$\frac{d^2}{(m+1)\,(a^2-b^2)}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-2}\,\left(a\,(n-2)+b\,(m+1)\,\mathrm{Sec}[e+f\,x]-a\,(m+n)\,\mathrm{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  a*d^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-2)/(f*(m+1)*(a^2-b^2)) -
  d^2/((m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-2)*(a*(n-2)+b*(m+1)*Csc[e+f*x]-a*(m+n)*Csc[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && LtQ[1,n,2] && IntegersQ[2*m,2*n]
```

3: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\land m<-1\land n>3$

Derivation: Nondegenerate secant recurrence 1a with $A\to c^2$, $B\to 2\,c\,d$, $C\to d^2$, $n\to n-2$, $p\to 0$

Rule: If $a^2-b^2\neq 0\land m<-1\land n>3$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx \longrightarrow$$
$$\frac{a^2\,d^3\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-3}}{b\,f\,(m+1)\,(a^2-b^2)}+$$
$$\frac{d^3}{b\,(m+1)\,(a^2-b^2)}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-3}\,\left(a^2\,(n-3)+a\,b\,(m+1)\,\mathrm{Sec}[e+f\,x]-\left(a^2\,(n-2)+b^2\,(m+1)\right)\mathrm{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -a^2*d^3*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-3)/(b*f*(m+1)*(a^2-b^2)) +
  d^3/(b*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-3)*
    Simp[a^2*(n-3)+a*b*(m+1)*Csc[e+f*x]-(a^2*(n-2)+b^2*(m+1))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && (IGtQ[n,3] || IntegersQ[n+1/2,2*m] && GtQ[n,2])
```

2. $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\wedge m<-1\wedge n\not>0$

1: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\wedge m+\frac{1}{2}\in\mathbb{Z}^-\wedge n\in\mathbb{Z}^-$

Derivation: Nondegenerate secant recurrence 1c with $A\to 1,\ B\to 0,\ C\to 0,\ p\to 0$

Rule: If $a^2-b^2\neq 0\wedge m+\frac{1}{2}\in\mathbb{Z}^-\wedge n\in\mathbb{Z}^-$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$
$$-\frac{\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^n}{a\,f\,n}-$$
$$\frac{1}{a\,d\,n}\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^{n+1}\,\left(b\,(m+n+1)-a\,(n+1)\,\mathrm{Sec}[e+f\,x]-b\,(m+n+2)\,\mathrm{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n/(a*f*n) -
  1/(a*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1)*
    Simp[b*(m+n+1)-a*(n+1)*Csc[e+f*x]-b*(m+n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && ILtQ[m+1/2,0] && ILtQ[n,0]
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\wedge m<-1\wedge n\not>0$

Derivation: Nondegenerate secant recurrence 1c with $A\to 1,\ B\to 0,\ C\to 0,\ p\to 0$

Rule: If $a^2-b^2\neq 0\wedge m<-1\wedge n\not>0$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$
$$-\frac{b^2\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^n}{a\,f\,(m+1)\,(a^2-b^2)}+$$

$$\frac{1}{a\,(m+1)\,\left(a^2-b^2\right)}\int (a+b\,\text{Sec}[e+f\,x])^{m+1}\,(d\,\text{Sec}[e+f\,x])^n\cdot$$
$$\left(a^2\,(m+1)-b^2\,(m+n+1)-a\,b\,(m+1)\,\text{Sec}[e+f\,x]+b^2\,(m+n+2)\,\text{Sec}[e+f\,x]^2\right)\,dx$$

–

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  b^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n/(a*f*(m+1)*(a^2-b^2)) +
  1/(a*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n*
    (a^2*(m+1)-b^2*(m+n+1)-a*b*(m+1)*Csc[e+f*x]+b^2*(m+n+2)*Csc[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,n},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && IntegersQ[2*m,2*n]
```

6. $\int \frac{(d\,\text{Sec}[e+f\,x])^n}{a+b\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0$

1. $\int \frac{(d\,\text{Sec}[e+f\,x])^n}{a+b\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\,\land\,n>0$

1: $\int \frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{a+b\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left(\sqrt{d\,\text{Cos}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}\right)==0$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{a+b\,\text{Sec}[e+f\,x]}\,dx\;\longrightarrow\;\frac{\sqrt{d\,\text{Cos}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}}{d}\int \frac{\sqrt{d\,\text{Cos}[e+f\,x]}}{b+a\,\text{Cos}[e+f\,x]}\,dx$$

–

Program code:

```
Int[Sqrt[d_.*csc[e_.+f_.*x_]]/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  Sqrt[d*Sin[e+f*x]]*Sqrt[d*Csc[e+f*x]]/d*Int[Sqrt[d*Sin[e+f*x]]/(b+a*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{(d\,\text{Sec}[e+f\,x])^{3/2}}{a+b\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left(\sqrt{d\,\text{Cos}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}\right)==0$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{(d\,\text{Sec}[e+f\,x])^{3/2}}{a+b\,\text{Sec}[e+f\,x]}\,dx \longrightarrow d\,\sqrt{d\,\text{Cos}[e+f\,x]}\,\sqrt{d\,\text{Sec}[e+f\,x]}\int \frac{1}{\sqrt{d\,\text{Cos}[e+f\,x]}\,(b+a\,\text{Cos}[e+f\,x])}\,dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^(3/2)/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  d*Sqrt[d*Sin[e+f*x]]*Sqrt[d*Csc[e+f*x]]*Int[1/(Sqrt[d*Sin[e+f*x]]*(b+a*Sin[e+f*x])),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

3: $\int \frac{(d \operatorname{Sec}[e+f x])^{5/2}}{a+b \operatorname{Sec}[e+f x]} \, dx$ when $a^2-b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{d z}{a+b z} == \frac{d}{b} - \frac{a d}{b (a+b z)}$

Rule: If $a^2-b^2 \neq 0$, then

$$\int \frac{(d \operatorname{Sec}[e+f x])^{5/2}}{a+b \operatorname{Sec}[e+f x]} \, dx \longrightarrow \frac{d}{b} \int (d \operatorname{Sec}[e+f x])^{3/2} \, dx - \frac{a d}{b} \int \frac{(d \operatorname{Sec}[e+f x])^{3/2}}{a+b \operatorname{Sec}[e+f x]} \, dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^(5/2)/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  d/b*Int[(d*Csc[e+f*x])^(3/2),x] - a*d/b*Int[(d*Csc[e+f*x])^(3/2)/(a+b*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

4: $\int \frac{(d\,\text{Sec}[e+f\,x])^n}{a+b\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\land n>3$

Derivation: Nondegenerate secant recurrence 1b with $A\to a^2$, $B\to 2\,a\,b$, $C\to b^2$, $m\to -3$, $p\to 0$

Rule: If $a^2-b^2\neq 0\land n>3$, then

$$\int \frac{(d\,\text{Sec}[e+f\,x])^n}{a+b\,\text{Sec}[e+f\,x]}\,dx \longrightarrow$$

$$\frac{d^3\,\text{Tan}[e+f\,x]\,(d\,\text{Sec}[e+f\,x])^{n-3}}{b\,f\,(n-2)}+\frac{d^3}{b\,(n-2)}\int \frac{(d\,\text{Sec}[e+f\,x])^{n-3}\left(a\,(n-3)+b\,(n-3)\,\text{Sec}[e+f\,x]-a\,(n-2)\,\text{Sec}[e+f\,x]^2\right)}{a+b\,\text{Sec}[e+f\,x]}\,dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -d^3*Cot[e+f*x]*(d*Csc[e+f*x])^(n-3)/(b*f*(n-2)) +
  d^3/(b*(n-2))*Int[(d*Csc[e+f*x])^(n-3)*Simp[a*(n-3)+b*(n-3)*Csc[e+f*x]-a*(n-2)*Csc[e+f*x]^2,x]/(a+b*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && GtQ[n,3]
```

2. $\int \frac{(d\,\text{Sec}[e+f\,x])^n}{a+b\,\text{Sec}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\land n<0$

1: $\int \frac{1}{\sqrt{d\,\text{Sec}[e+f\,x]}\,(a+b\,\text{Sec}[e+f\,x])}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{\sqrt{d\,z}\,(a+b\,z)}==\frac{b^2\,(d\,z)^{3/2}}{a^2\,d^2\,(a+b\,z)}+\frac{a-b\,z}{a^2\,\sqrt{d\,z}}$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{1}{\sqrt{d \operatorname{Sec}[e+f x]}\ (a+b \operatorname{Sec}[e+f x])} \, dx \to \frac{b^2}{a^2 d^2} \int \frac{(d \operatorname{Sec}[e+f x])^{3/2}}{a+b \operatorname{Sec}[e+f x]} \, dx + \frac{1}{a^2} \int \frac{a-b \operatorname{Sec}[e+f x]}{\sqrt{d \operatorname{Sec}[e+f x]}} \, dx$$

Program code:

```
Int[1/(Sqrt[d_.*csc[e_.+f_.*x_]]*(a_+b_.*csc[e_.+f_.*x_])),x_Symbol] :=
  b^2/(a^2*d^2)*Int[(d*Csc[e+f*x])^(3/2)/(a+b*Csc[e+f*x]),x] +
  1/a^2*Int[(a-b*Csc[e+f*x])/Sqrt[d*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{(d \operatorname{Sec}[e+f x])^n}{a+b \operatorname{Sec}[e+f x]} \, dx$ when $a^2-b^2 \neq 0 \land n \leq -1$

Derivation: Nondegenerate secant recurrence 1c with $A \to 1,\ B \to 0,\ C \to 0,\ p \to 0$

Rule: If $a^2-b^2 \neq 0 \land n \leq -1$, then

$$\int \frac{(d \operatorname{Sec}[e+f x])^n}{a+b \operatorname{Sec}[e+f x]} \, dx \to$$

$$-\frac{\operatorname{Tan}[e+f x]\ (d \operatorname{Sec}[e+f x])^n}{a f n} - \frac{1}{a d n} \int \frac{(d \operatorname{Sec}[e+f x])^{n+1} \left(b n - a (n+1) \operatorname{Sec}[e+f x] - b (n+1) \operatorname{Sec}[e+f x]^2\right)}{a+b \operatorname{Sec}[e+f x]} \, dx$$

–

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  Cot[e+f*x]*(d*Csc[e+f*x])^n/(a*f*n) -
  1/(a*d*n)*Int[(d*Csc[e+f*x])^(n+1)/(a+b*Csc[e+f*x])*
    Simp[b*n-a*(n+1)*Csc[e+f*x]-b*(n+1)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && LeQ[n,-1] && IntegerQ[2*n]
```

7. $\int \sqrt{a + b \operatorname{Sec}[e + f x]} \, (d \operatorname{Sec}[e + f x])^n \, dx$ when $a^2 - b^2 \neq 0$

1. $\int \sqrt{a + b \operatorname{Sec}[e + f x]} \, (d \operatorname{Sec}[e + f x])^n \, dx$ when $a^2 - b^2 \neq 0 \land n > 0$

1: $\int \sqrt{a + b \operatorname{Sec}[e + f x]} \, \sqrt{d \operatorname{Sec}[e + f x]} \, dx$ when $a^2 - b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\sqrt{a + b z} == \frac{a}{\sqrt{a+b z}} + \frac{b z}{\sqrt{a+b z}}$

Rule: If $a^2 - b^2 \neq 0$, then

$$\int \sqrt{a + b \operatorname{Sec}[e + f x]} \, \sqrt{d \operatorname{Sec}[e + f x]} \, dx \longrightarrow a \int \frac{\sqrt{d \operatorname{Sec}[e + f x]}}{\sqrt{a + b \operatorname{Sec}[e + f x]}} \, dx + \frac{b}{d} \int \frac{(d \operatorname{Sec}[e + f x])^{3/2}}{\sqrt{a + b \operatorname{Sec}[e + f x]}} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  a*Int[Sqrt[d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] +
  b/d*Int[(d*Csc[e+f*x])^(3/2)/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \sqrt{a + b \operatorname{Sec}[e + f x]} \, (d \operatorname{Sec}[e + f x])^n \, dx$ when $a^2 - b^2 \neq 0 \land n > 1$

Derivation: Secant recurrence 1b with $A \to 0,\ B \to 0,\ C \to 1,\ m \to m - 2,\ n \to \frac{1}{2}$

Derivation: Secant recurrence 3a with $A \to 0,\ B \to a,\ C \to b,\ m \to m - 1,\ n \to -\frac{1}{2}$

Rule: If $a^2 - b^2 \neq 0 \land n > 1$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\;(d\,\text{Sec}[e+f\,x])^n\,dx \longrightarrow$$

$$\frac{2\,d\,\text{Sin}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\;(d\,\text{Sec}[e+f\,x])^{n-1}}{f\,(2\,n-1)}+$$

$$\frac{d^2}{2\,n-1}\int\frac{(d\,\text{Sec}[e+f\,x])^{n-2}\,\left(2\,a\,(n-2)+b\,(2\,n-3)\,\text{Sec}[e+f\,x]+a\,\text{Sec}[e+f\,x]^2\right)}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$$

–

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -2*d*Cos[e+f*x]*Sqrt[a+b*Csc[e+f*x]]*(d*Csc[e+f*x])^(n-1)/(f*(2*n-1)) +
  d^2/(2*n-1)*Int[(d*Csc[e+f*x])^(n-2)*Simp[2*a*(n-2)+b*(2*n-3)*Csc[e+f*x]+a*Csc[e+f*x]^2,x]/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && GtQ[n,1] && IntegerQ[2*n]
```

2. $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\ (d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\land n<0$

1: $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{d\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Piecewise constant extraction

Basis: If $\partial_x \frac{\sqrt{a+b\,f[x]}}{\sqrt{d\,f[x]}\ \sqrt{b+a/f[x]}} == 0$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{d\,\text{Sec}[e+f\,x]}}\,dx \longrightarrow \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{d\,\text{Sec}[e+f\,x]}\ \sqrt{b+a\,\text{Cos}[e+f\,x]}}\int \sqrt{b+a\,\text{Cos}[e+f\,x]}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]/Sqrt[d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  Sqrt[a+b*Csc[e+f*x]]/(Sqrt[d*Csc[e+f*x]]*Sqrt[b+a*Sin[e+f*x]])*Int[Sqrt[b+a*Sin[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\ (d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\land n\leq -1$

Derivation: Nondegenerate secant recurrence 1a with $A\to 1,\ B\to 0,\ C\to 0,\ p\to 0$

Derivation: Nondegenerate secant recurrence 1c with $A\to c,\ B\to d,\ C\to 0,\ n\to n-1,\ p\to 0$

Rule: If $a^2-b^2\neq 0\land n\leq -1$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\ (d\,\text{Sec}[e+f\,x])^n\,dx \longrightarrow$$

$$-\frac{\text{Tan}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\left(d\,\text{Sec}[e+f\,x]\right)^n}{f\,n}-\frac{1}{2\,d\,n}\int\frac{\left(d\,\text{Sec}[e+f\,x]\right)^{n+1}\left(b-2\,a\,(n+1)\,\text{Sec}[e+f\,x]-b\,(2\,n+3)\,\text{Sec}[e+f\,x]^2\right)}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Cot[e+f*x]*Sqrt[a+b*Csc[e+f*x]]*(d*Csc[e+f*x])^n/(f*n) -
  1/(2*d*n)*Int[(d*Csc[e+f*x])^(n+1)*Simp[b-2*a*(n+1)*Csc[e+f*x]-b*(2*n+3)*Csc[e+f*x]^2,x]/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && LeQ[n,-1] && IntegerQ[2*n]
```

8. $\int \frac{(d \operatorname{Sec}[e+f x])^n}{\sqrt{a+b \operatorname{Sec}[e+f x]}} \, dx$ when $a^2-b^2 \neq 0$

1. $\int \frac{(d \operatorname{Sec}[e+f x])^n}{\sqrt{a+b \operatorname{Sec}[e+f x]}} \, dx$ when $a^2-b^2 \neq 0 \land n>0$

1: $\int \frac{\sqrt{d \operatorname{Sec}[e+f x]}}{\sqrt{a+b \operatorname{Sec}[e+f x]}} \, dx$ when $a^2-b^2 \neq 0$

- Derivation: Piecewise constant extraction

- Basis: If $\partial_x \frac{\sqrt{d\, f[x]}\ \sqrt{b+a\, f[x]^{-1}}}{\sqrt{a+b\, f[x]}} == 0$

- Rule: If $a^2-b^2 \neq 0$, then

$$\int \frac{\sqrt{d \operatorname{Sec}[e+f x]}}{\sqrt{a+b \operatorname{Sec}[e+f x]}} \, dx \to \frac{\sqrt{d \operatorname{Sec}[e+f x]}\ \sqrt{b+a \operatorname{Cos}[e+f x]}}{\sqrt{a+b \operatorname{Sec}[e+f x]}} \int \frac{1}{\sqrt{b+a \operatorname{Cos}[e+f x]}} \, dx$$

- Program code:

```
Int[Sqrt[d_.*csc[e_.+f_.*x_]]/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  Sqrt[d*Csc[e+f*x]]*Sqrt[b+a*Sin[e+f*x]]/Sqrt[a+b*Csc[e+f*x]]*Int[1/Sqrt[b+a*Sin[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

2. $\int \frac{(d\,\text{Sec}[e+f\,x])^n}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0\land n>1$

1: $\int \frac{(d\,\text{Sec}[e+f\,x])^{3/2}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Piecewise constant extraction

Basis: If $\partial_x \frac{\sqrt{d\,f[x]}\,\sqrt{b+a/f[x]}}{\sqrt{a+b\,f[x]}} == 0$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{(d\,\text{Sec}[e+f\,x])^{3/2}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx \longrightarrow \frac{d\,\sqrt{d\,\text{Sec}[e+f\,x]}\,\sqrt{b+a\,\text{Cos}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\int \frac{1}{\text{Cos}[e+f\,x]\,\sqrt{b+a\,\text{Cos}[e+f\,x]}}\,dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^(3/2)/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  d*Sqrt[d*Csc[e+f*x]]*Sqrt[b+a*Sin[e+f*x]]/Sqrt[a+b*Csc[e+f*x]]*Int[1/(Sin[e+f*x]*Sqrt[b+a*Sin[e+f*x]]),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{(d\,\text{Sec}[e+f\,x])^n}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0\land n>2$

Derivation: Secant recurrence 3a with $A\to 0,\ B\to 0,\ C\to 1,\ m\to m-2,\ n\to -\frac{1}{2}$

Rule: If $a^2-b^2\neq 0\land n>2$, then

$$\int \frac{(d\,\text{Sec}[e+f\,x])^n}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx \longrightarrow$$

$$\frac{2\,d^2\,\mathrm{Sin}[e+f\,x]\,(d\,\mathrm{Sec}[e+f\,x])^{n-2}\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}{b\,f\,(2\,n-3)}+$$

$$\frac{d^3}{b\,(2\,n-3)}\int\frac{1}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}(d\,\mathrm{Sec}[e+f\,x])^{n-3}\,\left(2\,a\,(n-3)+b\,(2\,n-5)\,\mathrm{Sec}[e+f\,x]-2\,a\,(n-2)\,\mathrm{Sec}[e+f\,x]^2\right)\,dx$$

–

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*d^2*Cos[e+f*x]*(d*Csc[e+f*x])^(n-2)*Sqrt[a+b*Csc[e+f*x]]/(b*f*(2*n-3)) +
  d^3/(b*(2*n-3))*Int[(d*Csc[e+f*x])^(n-3)/Sqrt[a+b*Csc[e+f*x]]*
     Simp[2*a*(n-3)+b*(2*n-5)*Csc[e+f*x]-2*a*(n-2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && GtQ[n,2] && IntegerQ[2*n]
```

2. $\int\frac{(d\,\mathrm{Sec}[e+f\,x])^n}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0\,\land\,n<0$

1: $\int\frac{1}{\mathrm{Sec}[e+f\,x]\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Nondegenerate secant recurrence 1c with $A\to 1,\ B\to 0,\ C\to 0,\ p\to 0$

–

Rule: If $a^2-b^2\neq 0$, then

$$\int\frac{1}{\mathrm{Sec}[e+f\,x]\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx\;\longrightarrow\;\frac{\mathrm{Sin}[e+f\,x]\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}{a\,f}-\frac{b}{2\,a}\int\frac{1+\mathrm{Sec}[e+f\,x]^2}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx$$

–

Program code:

```
Int[1/(csc[e_.+f_.*x_]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]),x_Symbol] :=
  -Cos[e+f*x]*Sqrt[a+b*Csc[e+f*x]]/(a*f) - b/(2*a)*Int[(1+Csc[e+f*x]^2)/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{1}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{d\,\mathrm{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{\sqrt{z}\,\sqrt{a+b\,z}} == \frac{\sqrt{a+b\,z}}{a\,\sqrt{z}} - \frac{b\,\sqrt{z}}{a\,\sqrt{a+b\,z}}$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{1}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{d\,\mathrm{Sec}[e+f\,x]}}\,dx \longrightarrow \frac{1}{a}\int \frac{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}{\sqrt{d\,\mathrm{Sec}[e+f\,x]}}\,dx - \frac{b}{a\,d}\int \frac{\sqrt{d\,\mathrm{Sec}[e+f\,x]}}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[d_.*csc[e_.+f_.*x_]]),x_Symbol] :=
  1/a*Int[Sqrt[a+b*Csc[e+f*x]]/Sqrt[d*Csc[e+f*x]],x] -
  b/(a*d)*Int[Sqrt[d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

3: $\int \frac{(d\,\text{Sec}[e+f\,x])^n}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0\land n<-1$

Derivation: Secant recurrence 3b with $A\to 1,\ B\to 0,\ C\to 0,\ n\to -\frac{1}{2}$

Rule: If $a^2-b^2\neq 0\land n<-1$, then

$$\int \frac{(d\,\text{Sec}[e+f\,x])^n}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx\longrightarrow$$

$$-\frac{\text{Sin}[e+f\,x]\,(d\,\text{Sec}[e+f\,x])^{n+1}\sqrt{a+b\,\text{Sec}[e+f\,x]}}{a\,d\,f\,n}+$$

$$\frac{1}{2\,a\,d\,n}\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}(d\,\text{Sec}[e+f\,x])^{n+1}\left(-b\,(2\,n+1)+2\,a\,(n+1)\,\text{Sec}[e+f\,x]+b\,(2\,n+3)\,\text{Sec}[e+f\,x]^2\right)dx$$

–

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  Cos[e+f*x]*(d*Csc[e+f*x])^(n+1)*Sqrt[a+b*Csc[e+f*x]]/(a*d*f*n) +
  1/(2*a*d*n)*Int[(d*Csc[e+f*x])^(n+1)/Sqrt[a+b*Csc[e+f*x]]*
    Simp[-b*(2*n+1)+2*a*(n+1)*Csc[e+f*x]+b*(2*n+3)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && LtQ[n,-1] && IntegerQ[2*n]
```

9: $\int (a+b\,\text{Sec}[e+f\,x])^{3/2}\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\land n\leq -1$

Derivation: Nondegenerate secant recurrence 1a with $A\to c,\ B\to d,\ C\to 0,\ n\to n-1,\ p\to 0$

–

Rule: If $a^2-b^2\neq 0\land n\leq -1$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^{3/2}\,(d\,\text{Sec}[e+f\,x])^n\,dx\longrightarrow$$

$$-\frac{a\,\mathrm{Tan}[e+f\,x]\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\left(d\,\mathrm{Sec}[e+f\,x]\right)^n}{f\,n}+$$

$$\frac{1}{2\,d\,n}\int\frac{1}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\left(d\,\mathrm{Sec}[e+f\,x]\right)^{n+1}\left(a\,b\,(2\,n-1)+2\left(b^2\,n+a^2\,(n+1)\right)\mathrm{Sec}[e+f\,x]+a\,b\,(2\,n+3)\,\mathrm{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^(3/2)*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  a*Cot[e+f*x]*Sqrt[a+b*Csc[e+f*x]]*(d*Csc[e+f*x])^n/(f*n) +
  1/(2*d*n)*Int[(d*Csc[e+f*x])^(n+1)/Sqrt[a+b*Csc[e+f*x]]*
     Simp[a*b*(2*n-1)+2*(b^2*n+a^2*(n+1))*Csc[e+f*x]+a*b*(2*n+3)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && LeQ[n,-1] && IntegersQ[2*n]
```

10: $\int\left(a+b\,\mathrm{Sec}[e+f\,x]\right)^m\left(d\,\mathrm{Sec}[e+f\,x]\right)^n dx$ when $a^2-b^2\neq 0\wedge n>3$

Derivation: Nondegenerate secant recurrence 1b with $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to m-2,\ p\to 0$

Rule: If $a^2-b^2\neq 0\wedge n>3$, then

$$\int\left(a+b\,\mathrm{Sec}[e+f\,x]\right)^m\left(d\,\mathrm{Sec}[e+f\,x]\right)^n dx\longrightarrow$$

$$\frac{d^3\,\mathrm{Tan}[e+f\,x]\left(a+b\,\mathrm{Sec}[e+f\,x]\right)^{m+1}\left(d\,\mathrm{Sec}[e+f\,x]\right)^{n-3}}{b\,f\,(m+n-1)}+$$

$$\frac{d^3}{b\,(m+n-1)}\int\left(a+b\,\mathrm{Sec}[e+f\,x]\right)^m\left(d\,\mathrm{Sec}[e+f\,x]\right)^{n-3}\left(a\,(n-3)+b\,(m+n-2)\,\mathrm{Sec}[e+f\,x]-a\,(n-2)\,\mathrm{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -d^3*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-3)/(b*f*(m+n-1)) +
  d^3/(b*(m+n-1))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-3)*
    Simp[a*(n-3)+b*(m+n-2)*Csc[e+f*x]-a*(n-2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,m},x] && NeQ[a^2-b^2,0] && GtQ[n,3] && (IntegerQ[n] || IntegersQ[2*m,2*n]) && Not[IGtQ[m,2]]
```

11: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\wedge\,0<m<2\,\wedge\,0<n<3\,\wedge\,m+n-1\neq 0$

Derivation: Nondegenerate secant recurrence 1b with $A\to a\,c,\ B\to b\,c+a\,d,\ C\to b\,d,\ m\to m-1,\ n\to n-1,\ p\to 0$

Rule: If $a^2-b^2\neq 0\,\wedge\,0<m<2\,\wedge\,0<n<3\,\wedge\,m+n-1\neq 0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\frac{b\,d\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-1}\,(d\,\text{Sec}[e+f\,x])^{n-1}}{f\,(m+n-1)}+$$

$$\frac{d}{m+n-1}\int (a+b\,\text{Sec}[e+f\,x])^{m-2}\,(d\,\text{Sec}[e+f\,x])^{n-1}\,\left(a\,b\,(n-1)+\left(b^2\,(m+n-2)+a^2\,(m+n-1)\right)\text{Sec}[e+f\,x]+a\,b\,(2\,m+n-2)\,\text{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -b*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n-1)/(f*(m+n-1)) +
  d/(m+n-1)*Int[(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^(n-1)*
    Simp[a*b*(n-1)+(b^2*(m+n-2)+a^2*(m+n-1))*Csc[e+f*x]+a*b*(2*m+n-2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && LtQ[0,m,2] && LtQ[0,n,3] && NeQ[m+n-1,0] && (IntegerQ[m] || IntegersQ[2*m,2*n])
```

12: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\land\,-1<m<2\,\land\,1<n<3\,\land\,m+n-1\neq 0$

Derivation: Nondegenerate secant recurrence 1b with $A\to a\,c,\ B\to b\,c+a\,d,\ C\to b\,d,\ m\to m-1,\ n\to n-1,\ p\to 0$

Rule: If $a^2-b^2\neq 0\,\land\,-1<m<2\,\land\,1<n<3\,\land\,m+n-1\neq 0$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,dx\longrightarrow$$

$$\frac{d^2\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^{n-2}}{f\,(m+n-1)}+$$

$$\frac{d^2}{b\,(m+n-1)}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m-1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-2}\,\left(a\,b\,(n-2)+b^2\,(m+n-2)\,\mathrm{Sec}[e+f\,x]+a\,b\,m\,\mathrm{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -d^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-2)/(f*(m+n-1)) +
  d^2/(b*(m+n-1))*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n-2)*
    Simp[a*b*(n-2)+b^2*(m+n-2)*Csc[e+f*x]+a*b*m*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && LtQ[-1,m,2] && LtQ[1,n,3] && NeQ[m+n-1,0] && (IntegerQ[n] || IntegersQ[2*m,2*n])
```

13: $\int \frac{(a+b\,\mathrm{Sec}[e+f\,x])^{3/2}}{\sqrt{d\,\mathrm{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{a+b\,z}{\sqrt{d\,z}} == \frac{a}{\sqrt{d\,z}} + \frac{b}{d}\sqrt{d\,z}$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{(a+b\,\mathrm{Sec}[e+f\,x])^{3/2}}{\sqrt{d\,\mathrm{Sec}[e+f\,x]}}\,dx \longrightarrow a\int \frac{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}{\sqrt{d\,\mathrm{Sec}[e+f\,x]}}\,dx + \frac{b}{d}\int \sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{d\,\mathrm{Sec}[e+f\,x]}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^(3/2)/Sqrt[d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  a*Int[Sqrt[a+b*Csc[e+f*x]]/Sqrt[d*Csc[e+f*x]],x] +
  b/d*Int[Sqrt[a+b*Csc[e+f*x]]*Sqrt[d*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

14: $\int (d\,\mathrm{Sec}[e+f\,x])^n\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\land m\in\mathbb{Z}$

Derivation: Piecewise constant extraction and algebraic simplification

Basis: $\partial_x\,(\mathrm{Cos}[e+f\,x]^n\,(d\,\mathrm{Sec}[e+f\,x])^n) == 0$

Rule: If $a^2-b^2\neq 0\land m\in\mathbb{Z}$, then

$$\int (d\,\mathrm{Sec}[e+f\,x])^n\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,dx \longrightarrow \mathrm{Cos}[e+f\,x]^n\,(d\,\mathrm{Sec}[e+f\,x])^n\int \frac{(a+b\,\mathrm{Sec}[e+f\,x])^m}{\mathrm{Cos}[e+f\,x]^n}\,dx$$

$$\longrightarrow \operatorname{Cos}[e+f x]^{n} (d \operatorname{Sec}[e+f x])^{n} \int \frac{(b+a \operatorname{Cos}[e+f x])^{m}}{\operatorname{Cos}[e+f x]^{m+n}} \, dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_.*(a_+b_.*csc[e_.+f_.*x_])^m_.,x_Symbol] :=
  Sin[e+f*x]^n*(d*Csc[e+f*x])^n*Int[(b+a*Sin[e+f*x])^m/Sin[e+f*x]^(m+n),x] /;
FreeQ[{a,b,d,e,f,n},x] && NeQ[a^2-b^2,0] && IntegerQ[m]
```

U: $\int (a+b \operatorname{Sec}[e+f x])^{m} (d \operatorname{Sec}[e+f x])^{n} \, dx$

Rule:

$$\int (a+b \operatorname{Sec}[e+f x])^{m} (d \operatorname{Sec}[e+f x])^{n} \, dx \longrightarrow \int (a+b \operatorname{Sec}[e+f x])^{m} (d \operatorname{Sec}[e+f x])^{n} \, dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,m,n},x]
```

Rules for integrands of the form $(d \operatorname{Cos}[e+f x])^m (a+b \operatorname{Sec}[e+f x])^p$

1: $\int (d \operatorname{Cos}[e+f x])^m (a+b \operatorname{Sec}[e+f x])^p \, dx$ when $m \notin \mathbb{Z} \wedge p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \left((d \operatorname{Cos}[e+f x])^m \left(\frac{\operatorname{Sec}[e+f x]}{d}\right)^m \right) == 0$

Rule: If $m \notin \mathbb{Z} \wedge p \notin \mathbb{Z}$, then

$$\int (d \operatorname{Cos}[e+f x])^m (a+b \operatorname{Sec}[e+f x])^p \, dx \rightarrow (d \operatorname{Cos}[e+f x])^{\operatorname{FracPart}[m]} \left(\frac{\operatorname{Sec}[e+f x]}{d}\right)^{\operatorname{FracPart}[m]} \int \left(\frac{\operatorname{Sec}[e+f x]}{d}\right)^{-m} (a+b \operatorname{Sec}[e+f x])^p \, dx$$

Program code:

```
Int[(d_.*cos[e_.+f_.*x_])^m_*(a_.+b_.*sec[e_.+f_.*x_])^p_,x_Symbol] :=
  (d*Cos[e+f*x])^FracPart[m]*(Sec[e+f*x]/d)^FracPart[m]*Int[(Sec[e+f*x]/d)^(-m)*(a+b*Sec[e+f*x])^p,x] /;
FreeQ[{a,b,d,e,f,m,p},x] && Not[IntegerQ[m]] && Not[IntegerQ[p]]
```